# Maintainer: Sergej Pupykin <pupykin.s+arch@gmail.com>
# Maintainer: Charles Mauch <cmauch@gmail.com>

pkgname=perl-term-progressbar
pkgver=2.23
pkgrel=1
pkgdesc="Perl/CPAN Module Term::ProgressBar : Progress bar in just ASCII  using Term"
arch=("any")
url="https://search.cpan.org/dist/Term-ProgressBar"
license=("GPL" "PerlArtistic")
depends=("perl-class-methodmaker" "perl-term-readkey")
source=("https://search.cpan.org/CPAN/authors/id/M/MA/MANWAR/Term-ProgressBar-$pkgver.tar.gz")
sha256sums=('defc03fb9f4ac1c9df1359d312bff3c0865ddefbf3aba64cd42a69a86215d49d')

build() {
  cd "$srcdir"/Term-ProgressBar-$pkgver
  PERL_MM_USE_DEFAULT=1 perl Makefile.PL INSTALLDIRS=vendor
  make
}

package() {
  cd "$srcdir"/Term-ProgressBar-$pkgver
  make install DESTDIR="$pkgdir"
  find "$pkgdir" -name '.packlist' -delete
  find "$pkgdir" -name '*.pod' -delete
}
